



## Engine Exhaust Aftertreatment with a Transient Inlet Flow

APP-Auto-5 (v2.0) September 9, 2010

### Summary

This application note demonstrates how you can employ a user-defined subroutine to read in transient engine-out data to set dynamic inlet conditions for an engine-exhaust aftertreatment simulation.

### Project Description

In this application note, the engine-out conditions as a function of time are stored in a tab-delimited text file. A user-editable subroutine is provided that reads the contents of this file and extracts from it the time-dependent inlet composition, temperature, and instantaneous flow rate into a transient perfectly stirred reactor model. The reactor approximates a 3-way catalytic converter, designed to convert  $\text{NO}_x$ , CO, and un-burned hydrocarbons (UHCs) through catalytic surface reactions on a platinum/rhodium catalyst. The gas-phase chemistry is neglected due to the dominance of the surface conversion reactions. This sample user routine is already compiled and linked into the standard installation, such that it can be run from the CHEMKIN-PRO Interface without the need of a FORTRAN compiler. Modifying the sample user routine would require access to a compatible FORTRAN compiler on the computer where CHEMKIN-PRO is installed. With the pre-compiled routine, however, you can change the numerical values of the engine-out data that is read by the user routine.

### Project Setup

The chemistry set used in this application note describes several processes that occur on a platinum/rhodium “three-way” catalyst. These processes are: the oxidation of unburned bicarbonates (represented by  $\text{C}_3\text{H}_6$ ) on Pt, the reduction of NO on Pt, the reduction of NO on Rh, and the oxidation of CO on Rh. This reaction mechanism is based on the published work of Chatterjee, et al.<sup>1</sup> in 2001. The reaction mechanism was developed for a surface made of 75% Pt and 25% Rh, and should be considered valid only for this composition. In particular, the coverage-dependent activation energies have been scaled for that Pt/Rh ratio.

Gas-phase reactions are not included in this chemistry set, although gas-phase species are created and destroyed at the catalytic surface. The Gas-phase Kinetics input file, therefore, only includes 6 elements: O, H, C, N, Rh and Pt and 9 gas-phase species:  $\text{O}_2$ ,  $\text{C}_3\text{H}_6$ ,  $\text{H}_2$ ,  $\text{H}_2\text{O}$ ,  $\text{CO}_2$ , CO, NO,  $\text{NO}_2$ , and  $\text{N}_2$ , but no reactions.

<sup>1</sup>D. Chatterjee, O. Deutschmann and J. Warnatz, “Detailed Surface Reaction Mechanism in a Three-way Catalyst”, *Faraday Discussions*, **119**:371-384 (2001).

The Surface Kinetics input file describes chemistry occurring on one material called 3WAYCATALYST that has two site types, each representing one of the metals, called PLATINUM and RHODIUM. The PLATINUM site can be occupied by any of 17 adsorbed surface (SITE) species, with Pt(S) representing an open site. One of the possible adsorbed species, the C<sub>3</sub>H<sub>6</sub>(S) species, actually occupies two sites. The RHODIUM site has simpler chemistry with only 5 surface-site species, with Rh(S1) representing the open site for this phase of the material. Thermo-chemical data are provided for some of the surface species, but others have place-holder values. All surface reactions are irreversible, such that none of the thermo-chemical data are used to obtain rates for reverse reactions. The PLATINUM site in this reaction mechanism has a site density of 2.04E-9 moles • cm<sup>-2</sup>, which is close to the number of 2.717E-9 moles • cm<sup>-2</sup> that is obtained from the density and molecular weight of solid platinum. This suggests that the creators of this mechanism may have chosen to use an empirically derived surface site density for the active rare-metal sites on the catalyst. This aspect should be kept in mind in applying this mechanism to other systems.

The oxidation of C<sub>3</sub>H<sub>6</sub> (or UHCs) on platinum is described in 47 irreversible reactions that include simple and dissociative adsorptions, simple and associative desorptions, plus reactions between adsorbed species. The latter include decomposition of adsorbed hydrocarbon species by hydrogen transfer to Pt(S) species, as well as various oxidation reactions for adsorbed hydrocarbon species ranging from a global description of C<sub>3</sub>H<sub>5</sub>(S) oxidation to more step-wise oxidation reactions for other hydrocarbon fragments. NO reduction on platinum is described in 5 reactions, while NO reduction and CO oxidation on rhodium are described in another 9 reactions, each involving adsorptions, desorptions and reactions among adsorbed species. Some reactions are described in terms of sticking coefficients, a few reactions have reaction-order overrides, and in several cases, the activation energy for a reaction varies quite strongly with the extent of coverage of the surface by one or more species.

The reactor diagram contains only one gas inlet and one perfectly-stirred reactor model (PSR). This diagram is the simplest way to represent the catalytic converter and is used here for demonstration purposes. This same approach, however, can be extended to account for axial gradients along the converter by representing the converter with a series of PSRs rather than a single PSR. When a series of PSRs are used, heat-transfer between the reactors in series can also be included, using the reactor-network heat-transfer connection option.

## Reactor and Inlet Settings

On the Reactor Physical Properties tab of the C1\_ PSR panel, first the problem type is selected as Solve Gas Energy Equation while selecting the Transient Solver. In this case, the reactor volume is the void volume in the converter honeycomb or porous media, estimated here as 1400 cm<sup>3</sup> for the entire length of the converter. Similarly, the reactive (internal) surface area within the converter is estimated based on the void geometry to be 59000 cm<sup>2</sup>. No residence time is input because the volume and flow rates are specified. The initial temperature (296.15 K) is input, along with the pressure (1 atm), and the fact that the system is treated as adiabatic (no heat loss).

On the Species-specific Properties tab of the C1\_ PSR panel, the initial conditions of the converter system, prior to the exhaust gas flowing through the system, are assumed to be air and are input on the Initial Gas Fraction sub-tab. The surface site fractions are estimated based on understanding of the surface conditions or on initial tests of the mechanism. A user-supplied initial guess can sometimes aid convergence; if it is not specified CHEMKIN-PRO will assume a uniform distribution of sites. In this case, we specify O(S) as the dominant site at the start for the platinum portion of the catalyst and CO(S1) for the rhodium portion. These site species are set to 1.0 on the Surface Fraction sub-tab to provide the initial conditions for the catalyst surface.

The gas inlet has been given a name that reflects its function, **engineout**. On the Stream Properties Data tab of the engineout panel, the Use Inlet User Routine option is selected, which indicates that information about the gas inlet composition, flow rate and temperatures should be obtained from the user inlet routine. The user routine pre-packaged with CHEMKIN-PRO reads a text file. The name of this file, **engineout.txt**, is set in the FORTRAN subroutine. The inlet conditions, some of which are shown in Table 1, are representative of measurements that might be taken during an engine test, where engine load and therefore exhaust flow rates and composition vary as a function of time. In this case, mole fractions for only a few species are provided (CO, NO, UHCs, and O<sub>2</sub>). We will assume that the balance of the gas can be represented by N<sub>2</sub> and that C<sub>3</sub>H<sub>6</sub> will chemically represent the UHCs. On this panel, the box is also checked to indicate that flow rates are given in volumetric flow units, rather than mass flow units.

Table 1 Excerpt of Data Representing Engine-out Test Measurements

<i>Time(s)</i>	<i>InletT(C)</i>	<i>Flow(SLM)</i>	<i>C<sub>3</sub>xx(ppm)</i>	<i>CO(ppm)</i>	<i>CO<sub>2</sub>(ppm)</i>	<i>NO<sub>x</sub>(ppm)</i>	<i>O<sub>2</sub>(ppm)</i>
<b>0</b>	23.9	87.08	924	4341	22764	38	127856
<b>0.5</b>	25	85.49	1238	23557	49535	127	77187
<b>1</b>	20.1	27.15	1369	36212	66499	126	50914
<b>1.5</b>	22.6	81.89	1267	37984	80793	118	28314
<b>2</b>	39.1	81.2	1176	29212	95735	92	14359
<b>2.5</b>	66.2	83.17	1016	18119	104492	89	9577
<b>3</b>	88.2	86	999	10391	110148	108	7744
<b>3.5</b>	111.3	87.08	912	5786	111276	101	13012

<i>Time(s)</i>	<i>InletT(C)</i>	<i>Flow(SLM)</i>	<i>C<sub>3xx</sub>(ppm)</i>	<i>CO(ppm)</i>	<i>CO<sub>2</sub>(ppm)</i>	<i>NO<sub>x</sub>(ppm)</i>	<i>O<sub>2</sub>(ppm)</i>
<b>4</b>	125.4	85.42	877	3663	112346	111	8645
<b>4.5</b>	141.2	82.53	843	2532	111864	104	11867
<b>5</b>	163	81.1	808	1967	111803	110	16508
<b>5.5</b>	175.2	82.43	681	1619	111685	101	12255
<b>6</b>	189.8	85.28	756	2224	110433	119	15728
<b>6.5</b>	208.9	87.06	825	6022	108765	207	7865
<b>7</b>	218.2	86.12	825	11532	107414	366	12835
<b>7.5</b>	230.6	83.34	732	9837	105707	440	20696
<b>8</b>	244.3	81.27	857	5259	102752	446	24211
<b>8.5</b>	251.6	193	957	2823	100132	381	28629
<b>9</b>	272.1	374.28	991	2942	99228	354	28534
<b>9.5</b>	297.4	479.67	864	2355	101356	331	28906
<b>10</b>	313.6	483.61	851	1514	103822	309	29511
<b>10.5</b>	333.2	431.87	796	1343	105570	276	23807
<b>11</b>	347	388.16	788	1251	106826	241	26600
<b>11.5</b>	349.6	344.94	868	910	106112	207	26409

### Details of the User Routine

The subroutine for the user-defined inlet is reproduced in Figure 1. The data in Table 1 has units of parts-per-million for species composition, standard liters per minute for flow rate, and degrees Celsius

for temperature, which are not the standard units used by the CHEMKIN-PRO software. The user routine must therefore perform necessary conversions before providing the data to the PSR program executable. It is also responsible for interpolating between data points for each time value needed during the time integration. At each time step during the transient simulation, the PSR program will call `USRINLET`. Parameters needed by the `USRINLET` subroutine are passed in through integer and real workspace arrays called `IINWRK` and `RINWRK`, respectively. The comments at the top of the code explain what parameters are stored in these arrays.

The sample `USRINLET` code in Figure 1 reads in data from a file on the first call, as determined when the time is equal to the starting time of the simulation. In this section, the routine uses the CHEMKIN-PRO library routine `CKCOMP` to find the location of a given species name in the gas-phase species array (`KNAMES`). Conversion from Celsius to Kelvin and from parts-per-million to mole fraction is performed as the data is read and stored on the first call. On subsequent calls, the data is accessed from memory (stored in a FORTRAN common block), and values are interpolated for the specified simulation time. The interpolation uses the CHEMKIN-PRO GAS-PHASE KINETICS library routine `CKBSEC` to linearly interpolate between the data points. Before returning, the routine converts the flow rate from standard liters per minute (slm) to standard cubic centimeters per minute (sccm). Note that the routine could have been written to return mass flow rate in units of g/sec. In this case, however, CHEMKIN-PRO Interface inputs tell the program to expect user-defined flow rates in units of sccm.

Figure 1 Sample USRINLET Subroutine for User-Defined Transient Inlet Conditions

```

SUBROUTINE USRINLET (LIUIN, IINWRK, LRUIN, RINWRK, INAME, KNAMES,
1 FLRT, TINL, TEIN, XIN)
!DEC$ IF DEFINED (DLLEXPORT)
!DEC$ ATTRIBUTES DLLEXPORT :: USRINLET
!DEC$ ENDIF
C This is a USER SUBROUTINE for defining Inlet properties
C as an arbitrary function of time.
C Use of this subroutine is controlled by the PSR program
C keyword, USRIN, as described in the CHEMKIN Input Manual.
C
C The subroutine USRINLET is used to supply values of
C FLRT - Real scalar, Mass flow rate in g/s
C (or if SCCM are the preferred units,
C use keyword LFPSC in addition to USRIN)
C TINL - Real scalar, Inlet temperature (K)
C TEIN - Real scalar, Inlet electron temperature (K)
C XIN(*) - Real array, Gas-phase reactant composition of
C the inlet (mole fraction);
C The length of this array is *exactly* KKGAS,
C the number of gas-phase species in the problem.
C and the user may set the Integer flag
C IINWRK(1) - if 0, inlet parameter setting was successful,
C else, there was a problem, so discontinue calculations
C
C Given the following data:
C
C LIUIN - Integer scalar, length of some Integer workspace
C IINWRK(*) - Integer workspace array, containing
C IINWRK(2), LOUT - if positive, the unit number of an open output file
C IINWRK(3), IPSR - PSR index number
C IINWRK(4), IINL - Index number of an Inlet of IPSR
C IINWRK(5), KKGAS- Gas-phase species count
C IINWRK(6), LENRGY-If 0, then inlet temperature TINL is fixed,
C else TINL must be supplied here
C IINWRK(7), LENRGE-If 0, then inlet electron temperature TEIN is fixed,
C else TEIN must be supplied here
C
C LRUIN - Integer scalar, length of some Real workspace
C RINWRK(*) - Real workspace array, containing
C RINWRK(1), TSTART - Initial time (sec.) of calculation
C RINWRK(2), TIM - Current time (sec.)
C
C INAME - Character-string Inlet name
C KNAMES- Character-string array, Gas-phase species names
C*****
IMPLICIT DOUBLE PRECISION (A-H, O-Z), INTEGER (I-N)
include 'user_routines_interface.inc'
PARAMETER (PATM = 1.01325D6, TSTD=273.15, ZERO=0.0D0)
C Variables passed in from calling routine
DIMENSION IINWRK(LIUIN), RINWRK(LRUIN)
CHARACTER*16 INAME, KNAMES(*)
C Arrays returned by this user routine
DIMENSION XIN(*)
C Local variables:
CHARACTER*80 MYFILE
PARAMETER (MYFILE = 'engineout.txt', LUNIT=33,
1 MXPTS=10000, MXSPEC=6)
C Local storage space for variables
CHARACTER*16 MYSPEC(MXSPEC)
CHARACTER*80 HEADER
LOGICAL LENRGY, LENRGE
DIMENSION PPM(MXSPEC)
COMMON/USRINL1/ MAPSP(MXSPEC), NPTS
COMMON/USRINL2/ TIMEPT(MXPTS), SLMPT(MXPTS), XINPT(MXPTS, MXSPEC),
1 TPT(MXPTS)
EXTERNAL CKUFIRST
SAVE IFIRST
DATA IFIRST/0/
C
C set error flag
IINWRK(1) = 0
C set local variables from workspace data provided
LOUT = IINWRK(2)
C acknowledge that USRINLET has been applied
IF (IFIRST .EQ. 0) CALL CKUFIRST (IFIRST, LOUT, 'USRINLET')
C
IPSR = IINWRK(3)
IINL = IINWRK(4)

```

```

KKGAS= IINWRK(5)
LENRGY=IINWRK(6).GT.0
LENRGE=IINWRK(7).GT.0
C
TSTART=RINWRK(1)
TIME =RINWRK(2)
C
C Initialize returned variables
FLRT = 0.0
TINL = 298.D0
TEIN = 298.D0
DO K = 1, KKGAS
  XIN(K) = 0.0D0
ENDDO
C
C First time in, read in all the points so we don't have to do
C IO on each call. Interpolate from saved points thereafter
IF (TIME .EQ. TSTART) THEN
C Open and read the time-date file
C Store points in arrays for access/interpolation at later times
IOS = 0
OPEN(LUNIT, FILE=MYFILE, FORM='FORMATTED', STATUS='OLD',
1 IOSTAT=IOS)
C Check that file open was successful, if not return with error
IF (IOS .NE. 0) GO TO 1000
C
C Map input species to CHEMKIN names and find indices
MYSPEC(1) = 'C3H6'
MYSPEC(2) = 'CO'
MYSPEC(3) = 'CO2'
MYSPEC(4) = 'NO'
MYSPEC(5) = 'O2'
MYSPEC(6) = 'N2'
DO MYK = 1, MXSPEC
  CALL CKCOMP(MYSPEC(MYK),KNAMES,KKGAS,INDX)
  IF (INDX .GT. 0 .AND. INDX .LE. KKGAS) THEN
    MAPSP(MYK) = INDX
  ENDDIF
ENDDO
C Read in the arrays of available information
C In this case file format is as follows:
C Time(s),T(C),Flrt(SLM), C3H6(ppm),CO(ppm),CO2(ppm),NO(ppm),O2(ppm)
READ(LUNIT, '(A)', END=800, ERR=1000) HEADER
NPTS = 0
DO I = 1, MXPTS
  READ(LUNIT, *, END=800, ERR=1000)
1 TIMEPT(I),TCELS,SLM,(PPM(K),K=1,MXSPEC-1)
  NPTS = NPTS + 1
  TPT(I) = TCELS + 273.15D0
  SLMPT(I) = SLM
  XSUM = 0.0D0
  DO MYK = 1, MXSPEC-1
    XINPT(I,MYK) = PPM(MYK) * 1.D-6
    XSUM = XSUM + XINPT(I,MYK)
  ENDDO
C Set fraction of N2 = 1 minus the sum of others
XREM = 1.0D0-XSUM
XINPT(I,MXSPEC) = MAX(XREM,ZERO)
C Note: if XSUM > 0.0, AURORA will normalize so that sum = 1
ENDDO
800 CONTINUE
IF (NPTS .EQ. 0) GO TO 1000
CLOSE(LUNIT)
ENDIF
C
C Interpolate data for input time and perform units conversions
XSUM = 0.0
DO MYK = 1, MXSPEC-1
  XIN(MAPSP(MYK)) = CKBSEC(NPTS,TIME,TIMEPT,XINPT(1,MYK))
  XSUM = XSUM + XIN(MAPSP(MYK))
ENDDO
XREM = 1.0-XSUM
XIN(MAPSP(MXSPEC)) = MAX(XREM, ZERO)
SLM = CKBSEC(NPTS,TIME,TIMEPT,SLMPT)
C Convert from SLM to SCCM
FLRT = SLM * 1000.D0
IF (LENRGY) THEN
C Set the gas inlet temperature
TINL = CKBSEC(NPTS,TIME,TIMEPT,TPT)
ENDIF
IF (LENRGE) THEN

```

```
C      Set the electron inlet temperature
      TEIN = TINL
      ENDIF
      RETURN
C
1000 CONTINUE
      IF (IOS .NE. 0) THEN
        WRITE (LOUT, *) ' ERROR...OPEN failure on inlet data file'
        CALL CKWARN(2)
      ELSE
        WRITE (LOUT, *) ' ERROR...READ failure on inlet data file'
        CALL CKWARN(2)
        CLOSE (LUNIT)
      ENDIF
      IINWRK(1) = 1
C
      RETURN
      END
```

## Project Results

Figure 2 shows molar conversions for  $C_3H_6$ , CO, NO as a function of time. The effectiveness of conversion can also be viewed by comparing the inlet mole fraction to the outlet mole fractions, as shown in Figure 3 for  $C_3H_6$ , or by looking at the calculated conversion efficiencies directly. These figures clearly show that  $C_3H_6$  and CO are converted more effectively than NO under these conditions. Early in the simulation, the conversion rates are low for all of these species. Although at  $t = 0$ , the calculated molar conversion is 100%, this is simply due to setting the initial conditions in the reactor to pure air, which determines the initial exit flow. At certain times the calculated conversion rates for CO and NO actually go negative. This results from the fact that CO and NO can be formed on the surface and thus can be “produced” as the state of the surface changes. Figure 4 shows inlet and exit gas temperatures as a function of time. Temperatures show that the gas heats up relative to the inlet gas due to exothermic surface reactions. The catalyst does not become effective until it reaches a temperature above about 600 K, which is consistent with the work reported by Chatterjee et al.<sup>2</sup>

<sup>2</sup> D. Chatterjee, O. Deutschmann and J. Warnatz, “Detailed Surface Reaction Mechanism in a Three-way Catalyst,” *Faraday Discussions*, **119**:371-384 (2001).

Figure 2 Engine Exhaust Aftertreatment—Molar Conversion Rates




Figure 3 Engine Exhaust Aftertreatment—Mole Fractions





Figure 4 Engine Exhaust Aftertreatment—Gas Temperatures



## About Reaction Design

Reaction Design helps transportation manufacturers and energy companies rapidly achieve their Clean Technology goals by automating the analysis of chemical processes via simulation and modeling solutions. Reaction Design is the exclusive developer and distributor of CHEMKIN and CHEMKIN-PRO, the *de facto* standards for modeling gas-phase and surface chemistry, providing engineers ultra-fast access to reliable answers that save time and money in the development process. Reaction Design also offers the CHEMKIN-CFD software module, which brings detailed kinetics modeling to other engineering applications, such as Computational Fluid Dynamics (CFD) programs. Reaction Design's world-class engineers, chemists and programmers have expertise that spans multi-scale engineering from the molecule to the production plant. Reaction Design serves more than 350 customers in the commercial, government and academic markets.

Reaction Design can be found online at [www.reactiondesign.com](http://www.reactiondesign.com).

*CHEMKIN® and Reaction Design® are registered trademarks of Reaction Design. CHEMKIN-PRO, CHEMKIN-CFD and Model Fuels Consortium are trademarks of Reaction Design.*